/*
 * Copyright the original author or authors.
 * 
 * Licensed under the MOZILLA PUBLIC LICENSE, Version 1.1 (the "License");
 * you may not use this file except in compliance with the License.
 * You may obtain a copy of the License at
 * 
 *	  http://www.mozilla.org/MPL/MPL-1.1.html
 * 
 * Unless required by applicable law or agreed to in writing, software
 * distributed under the License is distributed on an "AS IS" BASIS,
 * WITHOUT WARRANTIES OR CONDITIONS OF ANY KIND, either express or implied.
 * See the License for the specific language governing permissions and
 * limitations under the License.
 */
package com.googlecode.milib.service {
	import com.googlecode.milib.base.MLBroadcastClass;

	import flash.events.Event;
	import flash.net.SharedObject;
	import flash.utils.Dictionary;

	/**
	 * @author Marek Brun
	 */
	public class SharedObjectsObjectService extends MLBroadcastClass {
		
		private static const uniqueNameAndObject:Object={};

		private static const servicedObjects : Dictionary = new Dictionary(true);
		private var obj:ISharedObjectsServicedObject;
		private var cookie:SharedObject;
		
		public function SharedObjectsObjectService(access:Private, obj:ISharedObjectsServicedObject) {
			this.obj=obj;
			
			cookie=SharedObject.getLocal(obj.getSharedObjectUniqueName());
			if(!cookie.data.isInit){
				cookie.data.isInit=true;
				cookie.data.data=obj.getSharedObjectInitData();
				cookie.flush();
			}
			obj.applyCookieData(getSharedObjectData());
			obj.addEventListener(obj.event_SharedObjectDataChanged, onSharedObjectDataChanged);
		}
		
		public function getSharedObjectData():* {
			return cookie.data.data;
		}
		
		public static function forInstance(obj:ISharedObjectsServicedObject):SharedObjectsObjectService {
			if(!obj.getSharedObjectUniqueName()){
				throw Error('Object don\'t have name.');
			}
			if(uniqueNameAndObject[obj.getSharedObjectUniqueName()] && uniqueNameAndObject[obj.getSharedObjectUniqueName()]!=obj){
				throw Error('Object name ('+obj.getSharedObjectUniqueName()+') is not unique. Another object have alredy that name.');
			}
			if(servicedObjects[obj]){
				return servicedObjects[obj];
			}else{
				servicedObjects[obj]=new SharedObjectsObjectService(null, obj);
				return servicedObjects[obj];
			}
		}
		
//********************************************************************************************
//		events for Item
//********************************************************************************************
		protected function onSharedObjectDataChanged(event:Event):void {
			cookie.data.data=obj.getSharedObjectData();
			cookie.flush();
		}
		
	}
}

internal class Private {}

